Numerical Solution of the Lane-Emden Equation

Introduction

The numerical solution of differential equations in radial or polar coordinates can be complicated by the presence of a singularity at the origin (r=0). Bessel's equation is a well known example of this feature:

$$r^2 y'' + r y' + (r^2 - \nu^2) y = 0 \tag{1}$$

This equation has a regular singular point at r=0. The general solution of (1) when ν is an integer is

$$y(r) = c_1 J_\nu(r) + c_2 Y_\nu(r), \quad \nu = \text{integer} \tag{2}$$

Let us suppose that (1) must satisfy the following initial conditions for ν=0

$$\begin{aligned} \text{IC1}: &\quad y(0) = 1 \\ \text{IC2}: &\quad y'(0) = 0 \end{aligned} \tag{3}$$

The solution to this problem is

$$y(r) = J_0(r) \tag{4}$$

Let us attempt to solve this problem numerically. If we apply the boundary conditions at $r = 0$, NDSolve fails; it encounters a singularity at $r = 0$ and cannot start the integration.

```
sol = NDSolve[{r^2 y''[r] + r y'[r] + r^2 y[r] == 0, y[0] == 1, y'[0] == 0}, y[r], {r, 0, 10}]
```

Power::infy : Infinite expression $\frac{1}{0.}$ encountered. ≫

Infinity::indet : Indeterminate expression 0. ComplexInfinity encountered. ≫

NDSolve::ndnum : Encountered non−numerical value for a derivative at r == 0.`. ≫

```
NDSolve[{r^2 y[r] + r y′[r] + r^2 y″[r] == 0, y[0] == 1, y′[0] == 0}, y[r], {r, 0, 10}]
```

A simple solution to our dilemma is to apply the ICs at a finite value of r, say $r = 0.1$. In this case NDSolve avoids $r = 0$ and is able to start the integration:

```
sol =
 NDSolve[{r^2 y''[r] + r y'[r] + r^2 y[r] == 0, y[0.1] == 1, y'[0.1] == 0}, y[r], {r, 0.1, 20}]
{{y[r] → InterpolatingFunction[{{0.1, 20.}}, <>][r]}}
```

Here is a comparison plot of the numerical solution with the exact solution

```
Plot[{BesselJ[0, r], y[r] /.sol}, {r, 0.1, 20}, PlotStyle → Thick,
 Frame → True, FrameLabel → {Style["r", 16], Style["y(r)", 16]}]
```



From the plot it appears visually that we have excellent agreement but if we evaluate the solution at say $r = 15$ the numerical solution is only accurate to about 10%

```
{BesselJ[0, r], y[r] /. sol} /. r → 15.
```

```
{-0.0142245, {-0.01282}}
```

Even at r=5 the solution is only accurate to within 3% ! These calculations show that we need to be more selective in how we choose the ICs at $r = r_0$. In this example we apply the original IC conditions at $r = r_0$, and consequently the solution becomes less accurate for larger values of r.

An improved strategy is to develop a power series expansion for the solution about $r = 0$, and then use the series solution to determine the ICs at a finite value of r. We will illustrate this method by solving the Lane-Emden equation for a stellar structure. The reader should review the notebook ODESeriesSol-n.nb which shows how *Mathematica* can be used to determine the power series expansion of ODEs. Material and code from ODESeriesSoln.nb is used in this notebook.

Lane-Emden Equation for a Stellar Structure.

We begin our analysis by considering a fluid body that is sufficiently large and isolated that the body force acting on any element of the fluid is due to the gravitational attraction of its surrounding fluid elements. This is called a self-gravitating fluid body. The force **F** due to gravity acting on this body can be expressed as the gradient of a potential Ω

$$\mathbf{F} = -\nabla\Omega \tag{5}$$

where the gravitational potential Ω is given by Poisson's equation (see Appendix for derivation):

$$\nabla^2\Omega = 4\,\pi\,G\,\rho \tag{6}$$

In this equation G ($= 6.6726\times10^{-8}\ \mathrm{cm}^3\cdot\mathrm{g}^{-1}\cdot\mathrm{s}^{-2}$) is the gravitational constant and ρ is the fluid density which in general is a function of position. The condition for the fluid to be in hydrostatic equilibrium is given by (Batchelor, Ref1)

$$\nabla P = -\rho\,\nabla\Omega \tag{7}$$

Now suppose the fluid body has spherical symmetry (i.e., the density does not depend on the polar or azimuth directions), then Eqn. (7) reduces to a ODE in terms of the radial coordinate:

$$\frac{dP}{dr} = -\rho \frac{d\Omega}{dr} \tag{8}$$

Similarly, with spherical symmetry imposed, Eqn. (6) becomes

$$\frac{1}{r^2} \frac{d}{dr}\left(r^2 \frac{d\Omega}{dr}\right) = 4 \pi G\rho \tag{9}$$

If we integrate Eqn.(9) over r we get an expression for the radial force due to the gravitational field

$$F_r = \mathbf{F} \cdot \mathbf{e}_r = \frac{d\Omega}{dr} = \frac{G}{r^2} \int_0^r 4 \pi r^2 \rho \, dr = \frac{G M (r)}{r^2} \tag{10}$$

For the fluid body to be in hydrostatic equilibrium, the radial force must be balanced by the hydrostatic pressure gradient :

$$\frac{dP}{dr} = -\frac{G M (r) \rho (r)}{r^2} \tag{11}$$

where M(r) is the local mass of the fluid body defined as

$$M (r) = \int_0^r 4 \pi r^2 \rho \, dr \tag{12}$$

For analysis it convenient to express the hydrostatic pressure in terms of the density. This can be done by using Eqn.(8) to eliminate Ω from Eqn.(9) to get

$$\frac{1}{r^2} \frac{d}{d}\left(\frac{r^2}{\rho} \frac{dP}{dr}\right) = -4 \pi G\rho \tag{13}$$

To make further progress we need to have an equation of state that relates the hydrostatic pressure P to the fluid density ρ.

Historically, the equation of state used to describe a non-rotating star in hydrostatic equilibrium is the polytrope equation that relates pressure to star density

$$P = K \rho^{(1+1/n)} \tag{14}$$

The motivation for using (14) is based on the observation that convection within the star is so efficient that the gas behaves as an adiabatic gas. The quantities K and n are star constants, taken to be positive. It is not difficult to show that

$$\frac{d\rho^{1+1/n}}{dr} = \left(\frac{n+1}{n}\right) \rho^{\frac{1}{n}} \frac{d\rho}{dr} \tag{15}$$

Thus

$$\frac{1}{\rho} \frac{dP}{dr} = K \left(\frac{n+1}{n}\right) \rho^{-1+\frac{1}{n}} \frac{d\rho}{dr} \tag{16}$$

and

$$\frac{d\rho^{1/n}}{dr} = \frac{\rho^{-1+\frac{1}{n}}}{n} \frac{d\rho}{dr} \tag{17}$$

so that Eqn. (16) can be expressed as

$$\frac{1}{\rho} \frac{dP}{dr} = K (n+1) \frac{d\rho^{1/n}}{dr} \tag{18}$$

Using this result in Eqn. (13) gives

$$\frac{K\ (n+1)}{r^2}\ \frac{d}{d\xi}\left(r^2\ \frac{d\rho^{1/n}}{dr}\right) = -4\,\pi\,G\rho \qquad (19)$$

Next we introduce the following dimensionless variables:

$$\xi = \frac{r}{a},\quad \rho = \rho_c\,\phi^n \qquad (20)$$

The length scale a will be determined shortly, and ρ_c is the density at the center of the stellar structure. Using these variables, one can show that the equation for the star density becomes

$$\frac{1}{\xi^2}\ \frac{d}{d}\left(\xi^2\ \rho_c^{1/n}\ \frac{d\phi}{d\xi}\right) = -\frac{4\,\pi\,G\,\rho_c\,\phi^n}{K\ (n+1)} \qquad (21)$$

Regrouping terms gives

$$\frac{1}{\xi^2}\ \frac{d}{d}\left(\xi^2\ \frac{d\phi}{d\xi}\right) = -\frac{4\,\pi\,G\,\rho_c\,\phi^n\,a^2}{K\ (n+1)\ \rho_c^{1/n}} \qquad (22)$$

Thus if we define the length scale a to be

$$a = \sqrt{\frac{\rho_c^{1/n-1}\ K\ (n+1)}{4\,\pi\,G}} \qquad (23)$$

then we get the Lane-Emden equation for a stellar structure

$$\frac{1}{\xi^2}\ \frac{d}{d}\left(\xi^2\ \frac{d\phi}{d\xi}\right) + \phi^n = 0 \qquad (24)$$

The initial conditions for this equation are

IC1 : $\phi\ (0) = 1$

IC2 : $\phi'\ (0) = 0$

Chandrasekar (Ref 2) showed that one can find analytical solutions for (24)-(25) only if n has values $(0,\ 1,\ 5)$. The outer boundary of the star is defined by $\xi = \xi_0$ such that $\phi\ (\xi_0) = 0$, i.e., the radial location where the density of the star is zero.

In the next section we will describe a numerical scheme for finding the star size (value of ξ_0 when $\phi\ (\xi_0) = 0$) for different values of the polytropic index n.

Power Series Solution of Lane-Emden Equation

If we expand Eqn. (24) we see that the point ξ=0 represents a *regular singular point* of the ODE

$$\frac{d^2\phi}{d\xi^2} + \frac{2}{\xi}\ \frac{d\phi}{d\xi} + \phi^n = 0 \qquad (25)$$

Thus any numerical method will have difficulty starting at ξ=0. As discussed previously, an obvious way to get around this difficulty is to start the integration at some finite value of $\xi = \xi^*$. But then for accurate solutions, we are required to know what the appropriate ICs are at the new starting point. The approach we will take is to develop a power series expansion of ϕ about ξ=0. We then use this solution to determine the appropriate ICs at the new value of $\xi = \xi^*$.

We will make use of the following module that we developed in a previous notebook for finding power series solutions of ODEs about ξ=0:

```
ODESeriesSoln[sys_, indvar_, depvar_, n_] :=
 Module[{t = indvar, y = depvar, initconds, ODE, IC, PowerSeries,
   ODESeries, AlgebraicEqns, SeriesCoef, SeriesSol, plt1},
  initconds = Flatten[Sort[Select[Flatten[{sys}], D[#, t] &]]];
  ODE = Complement[Flatten[{sys}], initconds];
  IC = initconds /. Equal → Rule;
PowerSeries = Series[y[t], {t, 0, n}];
  ODESeries = First[ODE /. y[t] -> PowerSeries];
AlgebraicEqns = LogicalExpand[ODESeries] /. IC;
SeriesCoef = First[Solve[AlgebraicEqns]];
  SeriesSol = PowerSeries /. SeriesCoef /. IC;
SeriesSol]
```

Our system of equations is given by

$$\texttt{sys[n\_] :=} \left\{\frac{1}{\xi^2}\,\partial_\xi\left(\xi^2\,\partial_\xi\phi[\xi]\right)+\phi[\xi]^n == 0,\ \phi[0] == 1,\ \phi'[0] == 0\right\}$$

It will be convenient to define the following function that computes the series solution to $O\left(\xi^{13}\right)$

```
LESeries[n_] := ODESeriesSoln[sys[n], ξ, ϕ, 12]
```

Here is the series for the Lane-Emden equation about ξ=0 when n=2.

```
LESeries[2]
```

$$1-\frac{\xi^2}{6}+\frac{\xi^4}{60}-\frac{11\,\xi^6}{7560}+\frac{\xi^8}{8505}-\frac{97\,\xi^{10}}{10\,692\,000}+\frac{457\,\xi^{12}}{673\,596\,000}+O[\xi]^{13}$$

Here is the result for n=3.5

```
LESeries[7 / 2]
```

$$1-\frac{\xi^2}{6}+\frac{7\,\xi^4}{240}-\frac{23\,\xi^6}{4320}+\frac{77\,\xi^8}{77\,760}-\frac{637\,\xi^{10}}{3\,421\,440}+\frac{225\,799\,\xi^{12}}{6\,404\,935\,680}+O[\xi]^{13}$$

In the next section we use the series solution to construct initial conditions at $\xi \neq 0$

Numerical Solution with ICs applied at $\xi\neq 0$

We start our analysis by defining a function that is a list of our ODE with the new IC that depend on the value of ξ=ξ0, our starting value. Note also that since n can be a fractional power, we must anticipate how we are going to calculate ϕ^n. Since ϕ may become negative during the calculation, we write this term as `Sign (ϕ) Abs (ϕ)`^{n}. This will force $\phi^n$to be real number and not a complex number.

$$\texttt{sys2[n\_, }\xi\texttt{0\_] :=} \left\{\frac{1}{\xi^2}\,\partial_\xi\left(\xi^2\,\partial_\xi\phi[\xi]\right)+\mathrm{Sign}[\phi[\xi]]\,\mathrm{Abs}[\phi[\xi]]^n == 0,\right.$$
$$\left.\phi[\xi0] == \mathrm{Normal}[\mathrm{LESeries}[n]]\ /.\ \xi\to\xi0,\ \phi'[\xi0] == \mathrm{D}[\mathrm{Normal}[\mathrm{LESeries}[n]],\ \xi]\ /.\ \xi\to\xi0\right\}$$

The above system is a well-posed initial value problem which can be readily solved using NDSolve. Note we have increased the working precision of NDSolve:

```
sol = NDSolve[sys2[7 / 2, 1 / 10], ϕ[ξ], {ξ, 1 / 10, 10}, WorkingPrecision → 20]
```

```
{{ϕ[ξ] → InterpolatingFunction[
     {{0.10000000000000000000, 10.000000000000000000}}, <>][ξ]}}
```

Here is a plot of the solution when n=3.5

```
Plot[ϕ[ξ] /.sol, {ξ, 1/10, 10}, PlotStyle → Thick,
 Frame → True, FrameLabel → {Style["ξ", 16], Style["ϕ", 16]}]
```



We can determine the dimensionless size of the star when n=3.5 by finding the value of ξ when ϕ=0. We use FindRoot:

```
ξc = First[FindRoot[First[(ϕ[ξ] /. sol)] == 0, {ξ, 7, 10}, WorkingPrecision → 20]]; ξc
```

```
ξ → 9.5358053248749463333
```

The value listed by Chandrasekar is ξ=9.53581 for n=3.5. We can use the above code to determine star size for all possible values of n. The theory shows that for ***n>5*** there is no value of ξ such that ϕ=0!

Properties of Stars

In this section we will use the Lane-Emden analysis to estimate properties of our own Sun. The mass and radius of the Sun are

$$M_O = 1.99 \times 10^{30}\ \text{kg},\ \ R_O = 6.96 \times 10^{10}\ \text{cm} \tag{26}$$

Let us first attempt a crude analysis that uses Eqn. (11) to estimate the central pressure of the Sun

$$\frac{dP}{dr} \approx \frac{P\ (0) - P\ (R_O)}{(0 - R_O)} = -\frac{G\, M_O < \rho >}{R_O^2} \tag{27}$$

If we take the mean density of the Sun to be

$$< \rho > = \frac{M_O}{\frac{4}{3}\, \pi\, R_O^3} \tag{28}$$

Then the central pressure is

$$P_C = P(0) = \frac{3}{4\pi}\frac{G M_O^2}{R_O^4} = 3 \times 10^{15} \text{ dyn cm}^{-2} \tag{29}$$

This estimate is several orders to small! It has been shown that an appropriate polytrope to model the Sun is with n=3 (known as the Eddington Standard Model). We can solve the Lane-Emden equation with n=3 to get

```
sol = NDSolve[sys2[3, 1/10], φ[ξ], {ξ, 1/10, 10}, WorkingPrecision → 20];
Plot[{0, φ[ξ] /.sol}, {ξ, 1/10, 10},
 PlotStyle → {{Dashing[0.01], Red}, {Thick, Blue}}, Frame → True,
 FrameLabel → {Style["ξ", 16], Style["φ", 16]}, PlotRange → All, Axes → False]
```



The dimensionless star radius when n=3 is

```
ξc = First[FindRoot[First[(φ[ξ] /. sol)] == 0, {ξ, 7, 10}, WorkingPrecision → 20]];
ξc // InputForm
```

```
ξ -> 6.89684862714813509924100886307801144836`20.
```

This is in excellent agreement with the value given by Chandrasekar.

Let us now use the Lane-Emden model to determine the properties of our Sun. From Eqn. (12) the mass of the Sun is

$$M = \int_0^R 4\pi r^2 \rho \, dr = 4\pi a^3 \rho_c \int_0^{\xi_0} \xi^2 \phi^n \, d\xi \tag{30}$$

Integrating Eq. (24) gives

$$\xi_0^2 \left(\frac{d\phi}{d\xi}\right)_{\xi_0} = -\int_0^{\xi_0} \xi^2 \phi^n \, d\xi \tag{31}$$

Thus the mass of the star becomes

$$M = -4\pi a^3 \rho_c \xi_0^2 \left(\frac{d\phi}{d\xi}\right)_{\xi_0} \tag{32}$$

The mean density of the star is

$$\overline{\rho} = \frac{M}{\frac{4}{3}\pi R^3} = -\frac{3\,\rho_c}{\xi_0}\left(\frac{d\phi}{d\xi}\right)_{\xi_0} \tag{33}$$

Thus if we are given the mean density of the star we can compute ρ_c. Since we have the solution we can compute the terms involving ϕ in the RHS of (33). Using the values from our solution we have

```
-3/ξ ∂ξ (ϕ[ξ] /. sol) /. ξc
```

```
{0.0184561500}
```

Thus

$$\frac{\rho_c}{\overline{\rho}} = \frac{1}{0.018456150} = 54.1824 \tag{34}$$

Consider next the central pressure P_c of the star. From Eqn. (14) we can relate the central pressure to the central density

$$P_c = K\,\rho_c^{(1+1/n)} \tag{35}$$

From the definition of the length scale a we have

$$a^2 = \frac{\rho_c^{1/n-1}\,K\,(n+1)}{4\,\pi\,G} \tag{36}$$

Now using (36) to eliminate K from (37) we get the following expression for P_c

$$P_c = \frac{4\,\pi\,a^2\,G\,\rho_c^2}{(n+1)} \tag{37}$$

From (34) the central density of the star is

$$\rho_c = -\frac{M}{4\,\pi\,aR^2\,\left(\frac{d\phi}{d\xi}\right)} \tag{38}$$

Using this equation to eliminate $\rho_c$from (38) gives

$$P_c = \frac{M^2\,G}{4\,\pi\,R^4\,(n+1)\,\left(\frac{d\phi}{d\xi}\right)^2} \tag{39}$$

Thus if we know the mass of the star, its radius, and $d\phi / d\xi$ at $\xi = \xi_0$ we can compute the central pressure. For the Sun this calculation gives

```
M^2 G / (4 π R^4 (n + 1) (∂ξ (ϕ[ξ] /. sol))^2) /.
 {M → 1.99 × 10^33, R → 6.96 × 10^10, G → 6.6726 × 10^-8, n → 3, ξc}
```

$\{1.24438 \times 10^{17}\}$

Thus the central pressure of the Sun is $P_c = 1.2 \times 10^{17}$dyne cm^{-3} . We can also use (34) to determine the central density

$$\rho_c = \frac{M}{\frac{4}{3}\pi R^3}\,54.1824 \tag{40}$$

Evaluating (41) gives

$$\frac{\mathbf{M}}{\frac{4}{3}\pi \mathbf{R}^3} \mathbf{54.1824} \ /. \ \{\mathbf{M} \to \mathbf{1.99} \times \mathbf{10}^{33}, \mathbf{R} \to \mathbf{6.96} \times \mathbf{10}^{10}\}$$

76.3475

or $\rho_c = 76.347 \text{ g}/\text{cm}^3$. The length scale for the polytrope follows from (39) re-expressed as

$$a = \left(-\frac{M}{\rho_c \, 4\pi \, \xi_c^2 \left(\frac{d\phi}{d\xi}\right)_{\xi_c}}\right)^{1/3} \tag{41}$$

This evaluates to

$$\left(\left(-\frac{\mathbf{M}}{\rho_c \, \mathbf{4}\,\pi\,\xi^2 \, \partial_\xi \, (\phi[\xi] \ /. \ \mathbf{sol})}\right) /. \ \{\mathbf{M} \to \mathbf{1.99} \times \mathbf{10}^{33}, \rho_c \to \mathbf{76.3475}, \xi\mathbf{c}\}\right)^{1/3}$$

$\{1.00916 \times 10^{10}\}$

so that $a = 1.0092 \times 10^{10}$ cm. Finally we are interested in estimating the temperature of the Sun at its center. Stellar material is known to behave as like an ideal gas (even though the pressure and density are high). Hence the temperature is given by

$$T = \frac{P}{N\,k} \tag{42}$$

where N is the number density of particles is the gas plasma, and k is the Boltzman constant. N is comprised of electrons and ionized particles. A typical solar composition has a hyrogen mass fraction of ω=0.747, the remainder being mostly helium. If ρ is the local mass density of the star, then for a fully ionized gas each hydrogen atom contributes one electron and one charged hyrdogen ion so that the total number of hydrogen related particles in the plasma are

$$N_H = \frac{2\,\omega\,\rho}{m_p} \tag{43}$$

where $m_p$is the mass of a proton. Likewise, if helium is fully ionized, each helium atom contributes 2 electrons plus one charged helium ion to the stellar plasma. Since each helium ion has mass $4\,m_p$, then the number of helium related particles is

$$N_{He} = \frac{3\,(1-\omega)\,\rho}{4\,m_p} \tag{44}$$

It follows then that the number density of particles in the stellar plasma is

$$N = \frac{2\,\omega\,\rho}{m_p} + \frac{3\,(1-\omega)\,\rho}{4\,m_p} = \left(\frac{5\,\omega+3}{4}\right)\frac{\rho}{m_p} = \frac{\rho}{\mu\,m_p} \tag{45}$$

where $\mu \approx$0.6 is called *the mean molecular weight* of the stellar plasma.

We can express (43) in terms of the proton mass m_p:

$$T = \frac{P}{N\,k} = \frac{P\,\mu\,m_p}{\rho\,k} \tag{46}$$

Thus at the center of the star, the temperature is

$$T_c = \frac{P_c\, m_p\, \mu}{\rho_c\, k} \tag{47}$$

```
Pc mp μ / (k ρc) /. {Pc -> 1.244 × 10^17, ρc → 76.347, k → 1.3801 × 10^-16, mp → 1.672 × 10^-24, μ → 0.6}
```

```
1.18442 × 10^7
```

Thus our estimate of the central temperature of the Sun is $T_c = 11.84 \times 10^6$K

The most accurate model of our Sun is based on th Standard Solar Model (Bahcall, Physics Letters B, 433,1,1998). The central values based on that model are

$$T_c = 15.7 \times 10^6\ \mathrm{K}, \qquad P_c = 2.34 \times 10^{17}\ \mathrm{dynecm}^{-3} \qquad \rho_c = 152\ \mathrm{g}/\mathrm{cm}^3 \tag{48}$$

Thus the polytopic model based on th Lane-Emden model gives central values that are in reasonable agreement with the SSM of our Sun.

References

David Reies (http://www.scientificarts.com/laneemden/laneemden.html.) has an excellent notebook on the solution of the Lane-Emden equation using *Mathematica.* The approach he uses is essential the same as discussed here. In addition to his work , I found the following sources most helpful in compiling these notes.

- (1) G.K. Batchelor, *An Introduction to Fluid Dynamics,*Cambridge University Press,1967
- (2) S. Chandrasekar, *An Introduction to the Theory of Stellar Structure*, Dover, New York, 1958
- (3) Dennis G. Zill and Michael R. Cullen, *Advanced Engineering Mathematics*, PWS-KENT, Boston, MA, 1992

Appendix

If m and m' are two mass points, then Newton' inverse square law of gravitation is

$$\mathbf{F} = m\,m'\,G\,\frac{\mathbf{R}}{R^3} = m\,m'\,G\,\frac{\mathbf{e}_R}{R^2} \tag{49}$$

where G is the gravitational constant, and **R** is a position vector that defines the separation of the two point masses expressed in terms of relative coordinates

$$\mathbf{R} = \mathbf{r} - \mathbf{r}', \quad R = |\mathbf{r} - \mathbf{r}'|, \quad \mathbf{e}_r = \frac{\mathbf{R}}{R} \tag{50}$$

Now if m' is a distributed mass with density $\rho(\mathbf{r}')$, then the gravitational force is

$$\mathbf{F}(\mathbf{r}) = G\,m\int_{V'} \rho(\mathbf{r}')\,\frac{\mathbf{R}}{R^3}\,dV' \tag{51}$$

Thus the acceleration at the point $\mathbf{r}$ due to the distributed mass is

$$\mathbf{a}(\mathbf{r}) = G\int_{V'} \rho(\mathbf{r}')\,\frac{\mathbf{R}}{R^3}\,dV' \tag{52}$$

We will show next that the gradient of following scalar potential $\phi(\mathbf{r})$

$$\phi(\mathbf{r}) = -G \int_{V'} \rho(\mathbf{r}') \frac{1}{R} dV' \tag{53}$$

is equal to $-\mathbf{a}(\mathbf{r})$. To prove this result we will need to know how to differentiate vectors in terms of relative coordinates. Let us start be writing the position vectors $\mathbf{r}$ and $\mathbf{r}'$ and in terms of their components

$$\mathbf{r} = x\,\mathbf{e}_x + y\,\mathbf{e}_y + z\,\mathbf{e}_z, \quad \mathbf{r}' = x'\,\mathbf{e}_x + y'\,\mathbf{e}_y + z'\,\mathbf{e}_z \tag{54}$$

Thus

$$\mathbf{R} = (x - x')\,\mathbf{e}_x + (y - y')\,\mathbf{e}_y + (z - z')\,\mathbf{e}_z \tag{55}$$

and the magnitude of $\mathbf{R}$ is

$$R = |\mathbf{R}| = \sqrt{(x - x')^2 + (y - y')^2 + (z - z')^2} \tag{56}$$

Consider next the derivative of R with respect to x

$$\frac{\partial R}{\partial x} = \frac{(x - x')}{R} = -\frac{\partial R}{\partial x'} \tag{57}$$

so that

$$\nabla R = \frac{\mathbf{R}}{R} = \mathbf{e}_R = -\nabla' R \tag{58}$$

Also

$$\nabla\left(\frac{1}{R}\right) = -\frac{\mathbf{R}}{R^3}, \quad \nabla\cdot\nabla\left(\frac{1}{R}\right) = -\frac{3}{R^3} + \frac{3R^2}{R^5} = 0, \quad R \neq 0 \tag{59}$$

Thus if we take the gradient of ϕ with respect to the unprimed coordinates

$$\begin{aligned}\nabla\phi &= \nabla\left(-G \int_{V'} \rho(\mathbf{r}') \frac{1}{R} dV'\right) = -G \int_{V'} \rho(\mathbf{r}')\, \nabla\left(\frac{1}{R}\right) dV' \\ &= -G \int_{V'} \rho(\mathbf{r}') \frac{\mathbf{R}}{R^3} dV'\end{aligned} \tag{60}$$

Next we take the divergence of Eqn (60) with respect to the unprimed coordinates

$$\nabla\cdot\nabla\phi = \nabla^2\phi = -G \int_{V'} \rho(\mathbf{r}')\, \nabla\cdot\left(\frac{\mathbf{R}}{R^3}\right) dV' \tag{61}$$

For all points $\mathbf{r} \neq \mathbf{r}'$, we have from Eqn. (59) that $\nabla\cdot\left(\frac{\mathbf{R}}{R^3}\right) = 0$. Hence to evaluate this integral we need to make use of a limiting process. We consider a small spherical region about $\mathbf{r}'$ with volume $\triangle V'$ Since $\mathbf{r}' \approx \mathbf{r}$, we can assume that $\rho(\mathbf{r}')$ is approximately constant and equal to $\rho(\mathbf{r})$, which then can be taken out of the integral

$$\nabla^2\phi = -G\rho(\mathbf{r}) \int_{\triangle V'} \nabla\cdot\left(\frac{\mathbf{R}}{R^3}\right) dV' \tag{62}$$

The integrand can be evaluated if we transform to primed coordinates

$$\nabla^2\phi = G\rho(\mathbf{r}) \int_{\triangle V'} \nabla'\cdot\left(\frac{\mathbf{R}}{R^3}\right) dV' \tag{63}$$

Invoking the divergence theorem gives

$$\nabla^2 \phi = G \rho (\mathbf{r}) \int_{\Delta S'} \left(\frac{\mathbf{R}}{R^3} \right) \cdot \mathbf{e}_r \, dS' \tag{64}$$

where ΔS' is the surface area of our limiting spherical volume about r'. We can recognize the integrand in Eqn. (64) to be the element of solid angle $d\Omega$ subtended at **r** by the area dS' so that

$$\nabla^2 \phi = G \rho (\mathbf{r}) \oint d\Omega = 4 \pi G \rho (\mathbf{r}) \tag{65}$$

This is the Poisson equation for the gravitational potential ϕ.